

**Санкт-Петербургский государственный политехнический
университет**

Физико-механический
факультет

Кафедра
«Прикладная математика»

И з м а й л о в Г. К.

**Типовое экзаменационное задание по программированию
в рамках учебного курса «ИНФОРМАТИКА»**

Санкт-Петербург
2004 г.

*«Нахождение глубинной простоты
в запутанном клубке сущностей – это
и есть творчество в программировании».*

Х. Д. Милн

Представлен образец разработки проекта программы методом пошаговой детализации. При такой разработке алгоритм рассматривается как некоторый постоянно совершенствуемый объект, когда из начальной стандартной схемы «ввод-обработка-вывод» путем последовательных уточнений (детализаций) получаем вариант, непосредственно кодируемый на том или ином универсальном языке программирования, в данном случае на языке Turbo-Pascal.

Задания экзаменуемым таковы, чтобы при их программировании проверялись бы приобретенные в семестре навыки в обособлении некоторых подзадач в виде подпрограмм, использовании массивов в качестве параметров подпрограмм, а также такого важного элемента программирования, как ввод и вывод через файлы.

Задание

По заданным целым числам x и n , сформировать матрицу n -го порядка, следующим образом ($n \leq 20$)

$$A = \begin{matrix} 1 & x & x^2 & \dots & x^{n-1} \\ x & x^2 & x^3 & \dots & x^n \\ x^2 & x^3 & x^4 & \dots & x^n \\ \dots & \dots & \dots & \dots & \dots \\ x^{n-1} & x^n & \dots & \dots & x^{2n-2} \end{matrix}$$

Головной алгоритм

Алгоритм Form

объявление переменных
объявление процедуры (Creation)

начало

ввод исходных данных
обратиться к процедуре
вывести сформированную матрицу

конец

Подчиненный алгоритм

Начальный этап проектирования

Алгоритм Creation(цел массив A, цел x, n)

знач x, n
рез A
{объявление локальных переменных}

начало

заполнить матрицу A требуемыми элементами

конец

Первый этап детализации

Алгоритм Creation(цел массив A, цел x, n)

знач x, n
рез A
{объявление локальных переменных}

начало

сформировать первую строку матрицы A
для i от 2 до n выполнять
формировать i -ую строку

конец-цикл

конец

Окончательный вариант проекта подчиненного алгоритма

Алгоритм Creation(цел массив A, цел x, n)

знач x, n

рез A

цел i, j

начало

$a_{11} \leftarrow 1$

для j от 2 до n выполнять

$a_{1j} \leftarrow a_{1,j-1} * x$

конец-цикл

для i от 2 до n выполнять

для j от 1 до (n-1) выполнять

$a_{ij} \leftarrow a_{i-1,j+1}$



конец-цикл {j}

$a_{i,n} \leftarrow a_{i,n-1} * x$

конец-цикл {i}

конец {Creation}

Блок-схемы головного и подчиненного алгоритмов



Программа, написанная на Turbo Pascal.

```
Program exam_example;
uses crt;
const n=5 ;
type
  mas=array [1..n,1..n] of integer;
var
  x,i,j:integer;
  A:mas;
  fout : text;

procedure Creation (var A:mas;
                   x,n:integer);
var
  i,j:integer;
begin
  a[1,1]:=1;
  for j:=2 to n do
    a[1,j]:=a[1,j-1]*x;
  for i:=2 to n do
    begin
      for j:=1 to n-1 do
        a[i,j]:= a[i-1,j+1];
      a[i,n]:=a[i,n-1]*x;
    end;
end; {Creation}
begin
  clrscr;
  assign(fout,'exam.res');
  rewrite(fout);
  writeln('Введите x =>');
  readln(x);
  Creation(A,x,n);
  writeln(fout,'Для x=',x:3,' и n=',n:2,'сформирована матрица:');
  writeln;
  for i:=1 to n do
    begin
      for j:=1 to n do
        write(fout,' ',a[i,j]:4);
      writeln(fout);
    end;
  close(fout);
  writeln('Результат в файле Exam.res!');
  readln
end.
```

Тест

Пусть $x=2$; $n=5$.

Тогда формируемая матрица должна иметь вид

$$A = \begin{matrix} 1 & 2 & 4 & 8 & 16 \\ 2 & 4 & 8 & 16 & 32 \\ 4 & 8 & 16 & 32 & 64 \\ 16 & 32 & 64 & 128 & 256 \end{matrix}$$

Результаты работы программы совпадают с контрольными значениями.